**tssmooth ma —** Moving-average filter

**Description**

`tssmooth ma` creates a new series in which each observation is an average of nearby observations in the original series. The moving average may be calculated with uniform or user-specified weights. Missing periods are excluded from calculations.

**Quick start**

Create `ma1` using a second-order moving average of `y1` with `tsset` data

```
tssmooth ma ma1 = y1, window(2)
```

Also include the current observation in the average

```
tssmooth ma ma1 = y1, window(2 1)
```

Also include 4 forward terms in the average

```
tssmooth ma ma1 = y1, window(2 1 4)
```

Create `ma2` using a moving average of `y2` with weight 2 for the first lag of `y2`, 3 for its current value, 5 for its first forward value, and 4 for its second forward value

```
tssmooth ma ma2 = y2, weights(2 <3> 5 4)
```

Note: The above commands can also be used to apply the smoother separately to each panel of a panel dataset when a `panelvar` has been specified using `tsset` or `xtset`.

**Menu**

Statistics > Time series > Smoothers/univariate forecasters > Moving-average filter
Syntax

Moving average with uniform weights

tssmooth ma [type] newvar = exp [if] [in], window(#l[#c[#f]]) [replace]

Moving average with specified weights

tssmooth ma [type] newvar = exp [if] [in], weights([numlist] <#c> [numlist_f]) [replace]

You must tsset your data before using tssmooth ma; see [TS] tsset.

exp may contain time-series operators; see [U] 11.4.4 Time-series varlists.

Options

window(#l[#c[#f]]) describes the span of the uniformly weighted moving average.

#l specifies the number of lagged terms to be included, 0 ≤ #l ≤ one-half the number of observations in the sample.

#c is optional and specifies whether to include the current observation in the filter. A 0 indicates exclusion and 1, inclusion. The current observation is excluded by default.

#f is optional and specifies the number of forward terms to be included, 0 ≤ #f ≤ one-half the number of observations in the sample.

weights([numlist] <#c> [numlist_f]) is required for the weighted moving average and describes the span of the moving average, as well as the weights to be applied to each term in the average. The middle term literally is surrounded by < and >, so you might type weights(1/2 <3> 2/1).
	numlist_l is optional and specifies the weights to be applied to the lagged terms when computing the moving average.

#c is required and specifies the weight to be applied to the current term.

numlist_f is optional and specifies the weights to be applied to the forward terms when computing the moving average.

The number of elements in each numlist is limited to one-half the number of observations in the sample.

replace replaces newvar if it already exists.

Remarks and examples

stata.com

Remarks are presented under the following headings:

Overview
Video example
Overview

Moving averages are simple linear filters of the form

$$\hat{x}_t = \frac{\sum_{i=-l}^{f} w_i x_{t+i}}{\sum_{i=-l}^{f} w_i}$$

where

- $\hat{x}_t$ is the moving average
- $x_t$ is the variable or expression to be smoothed
- $w_i$ are the weights being applied to the terms in the filter
- $l$ is the longest lag in the span of the filter
- $f$ is the longest lead in the span of the filter

Moving averages are used primarily to reduce noise in time-series data. Using moving averages to isolate signals is problematic, however, because the moving averages themselves are serially correlated, even when the underlying data series is not. Still, Chatfield (2004) discusses moving-average filters and provides several specific moving-average filters for extracting certain trends.

Example 1: A symmetric moving-average filter with uniform weights

Suppose that we have a time series of sales data, and we want to separate the data into two components: signal and noise. To eliminate the noise, we apply a moving-average filter. In this example, we use a symmetric moving average with a span of 5. This means that we will average the first two lagged values, the current value, and the first two forward terms of the series, with each term in the average receiving a weight of 1.

```
. use https://www.stata-press.com/data/r16/sales1
. tsset
time variable:  t, 1 to 50
delta:  1 unit
. tssmooth ma sm1 = sales, window(2 1 2)
The smoother applied was
(1/5)*[x(t-2) + x(t-1) + 1*x(t) + x(t+1) + x(t+2)]; x(t)= sales
```

We would like to smooth our series so that there is no autocorrelation in the noise. Below we compute the noise as the difference between the smoothed series and the series itself. Then we use `ac` (see `[TS] corrgam`) to check for autocorrelation in the noise.
Example 2: A symmetric moving-average filter with nonuniform weights

In the previous example, there is some evidence of negative second-order autocorrelation, possibly due to the uniform weighting or the length of the filter. We are going to specify a shorter filter in which the weights decline as the observations get farther away from the current observation.

The weighted moving-average filter requires that we supply the weights to apply to each element with the *weights()* option. In specifying the weights, we implicitly specify the span of the filter.

Below we use the filter

\[
\hat{x}_t = \frac{1}{9}(x_{t-2} + 2x_{t-1} + 3x_t + 2x_{t+1} + x_{t+2})
\]

In what follows, 1/2 does not mean one-half, it means the numlist 1 2:

```
. tssmooth ma sm2 = sales, weights( 1/2 <3> 2/1)
The smoother applied was
   (1/9)*[1*x(t-2) + 2*x(t-1) + 3*x(t) + 2*x(t+1) + x(t+2)]; x(t)= sales
. generate noise2 = sales-sm2
```

We compute the noise and use *ac* to check for autocorrelation.
The graph shows no significant evidence of autocorrelation in the noise from the second filter.

Technical note

`tssmooth ma` gives any missing observations a coefficient of zero in both the uniformly weighted and weighted moving-average filters. This simply means that missing values or missing periods are excluded from the moving average.

Sample restrictions, via `if` and `in`, cause the expression smoothed by `tssmooth ma` to be missing for the excluded observations. Thus sample restrictions have the same effect as missing values in a variable that is filtered in the expression. Also, gaps in the data that are longer than the span of the filter will generate missing values in the filtered series.

Because the first \( l \) observations and the last \( f \) observations will be outside the span of the filter, those observations will be set to missing in the moving-average series.

Video example

Moving-average smoothers
tssmooth ma — Moving-average filter

### Stored results

`tssmooth ma` stores the following in `r()`:

**Scalars**
- `r(N)` number of observations
- `r(w0)` weight on the current observation
- `r(wlead#)` weight on lead #, if leads are specified
- `r(wlag#)` weight on lag #, if lags are specified

**Macros**
- `r(method)` smoothing method
- `r(exp)` expression specified
- `r(timevar)` time variable specified in `tsset`
- `r(panelvar)` panel variable specified in `tsset`

### Methods and formulas

The formula for moving averages is the same as previously given.

A truncated description of the specified moving-average filter labels the new variable. See [D] `label` for more information on labels.

An untruncated description of the specified moving-average filter is saved in the characteristic `tssmooth` for the new variable. See [P] `char` for more information on characteristics.

### Reference


### Also see

[Ts] `tsset` — Declare data to be time-series data
[Ts] `tssmooth` — Smooth and forecast univariate time-series data